

# NAG Fortran Library Routine Document

## F12ABF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F12ABF is an iterative solver in a suite of routines consisting of F12ABF, F12AAF, F12ACF, F12ADF and F12AEF. It is used to find some of the eigenvalues (and optionally the corresponding eigenvectors) of a standard or generalized eigenvalue problem defined by real nonsymmetric matrices.

### 2 Specification

```

SUBROUTINE F12ABF (IREVCM, RESID, V, LDV, X, MX, NSHIFT, COMM, ICOMM,
1                IFAIL)
    INTEGER          IREVCM, LDV, NSHIFT, ICOMM(*), IFAIL
    double precision RESID(*), V(LDV,*), X(*), MX(*), COMM(*)

```

### 3 Description

The suite of routines is designed to calculate some of the eigenvalues,  $\lambda$ , (and optionally the corresponding eigenvectors,  $x$ ) of a standard eigenvalue problem  $Ax = \lambda x$ , or of a generalized eigenvalue problem  $Ax = \lambda Bx$  of order  $n$ , where  $n$  is large and the coefficient matrices  $A$  and  $B$  are sparse, real and nonsymmetric. The suite can also be used to find selected eigenvalues/eigenvectors of smaller scale dense, real and nonsymmetric problems.

F12ABF is a **reverse communication** routine, based on the ARPACK routine **dnaupd**, using the Implicitly Restarted Arnoldi iteration method. The method is described in Lehoucq and Sorensen (1996) and Lehoucq (2001) while its use within the ARPACK software is described in great detail in Lehoucq *et al.* (1998). An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices is provided in Lehoucq and Scott (1996). This suite of routines offers the same functionality as the ARPACK software for real nonsymmetric problems, but the interface design is quite different in order to make the option setting clearer to the user and to simplify the interface of F12ABF.

The setup routine F12AAF must be called before F12ABF, the reverse communication iterative solver. Options may be set for F12ABF by prior calls to the option setting routine F12ADF and a post-processing routine F12ACF must be called following a successful final exit from F12ABF. F12AEF, may be called following certain flagged, intermediate exits from F12ABF to provide additional monitoring information about the computation.

F12ABF uses **reverse communication**, i.e., it returns repeatedly to the calling program with the parameter IREVCM (see Section 5) set to specified values which require the calling program to carry out one of the following tasks:

- compute the matrix-vector product  $y = OPx$ , where  $OP$  is defined by the computational mode;
- compute the matrix-vector product  $y = Bx$ ;
- notify the completion of the computation;
- allow the calling program to monitor the solution.

The problem type to be solved (standard or generalized), the spectrum of eigenvalues of interest, the mode used (regular, regular inverse, shifted inverse, shifted real or shifted imaginary) and other options can all be set using the option setting routine F12ADF (see Section 10.2 of the document for F12ADF for details on setting options and of the default settings).

## 4 References

Lehoucq R B (2001) Implicitly Restarted Arnoldi Methods and Subspace Iteration *SIAM Journal on Matrix Analysis and Applications* **23** 551–562

Lehoucq R B and Scott J A (1996) An evaluation of software for computing eigenvalues of sparse nonsymmetric matrices *Preprint MCS-P547-1195* Argonne National Laboratory

Lehoucq R B and Sorensen D C (1996) Deflation Techniques for an Implicitly Restarted Arnoldi Iteration *SIAM Journal on Matrix Analysis and Applications* **17** 789–821

Lehoucq R B, Sorensen D C and Yang C (1998) *ARPACK Users' Guide: Solution of Large-Scale Eigenvalue Problems with Implicitly Restarted Arnoldi Methods* SIAM, Philadelphia

## 5 Parameters

**Note:** this routine uses **reverse communication**. Its use involves an initial entry, intermediate exits and re-entries, and a final exit, as indicated by the **parameter IREVCM**. Between intermediate exits and re-entries, **all parameters other than IREVCM, RESID and V must remain unchanged**.

1: IREVCM – INTEGER *Input/Output*

*On initial entry:* IREVCM = 0, otherwise an error condition will be raised.

*On intermediate re-entry:* IREVCM must be unchanged from its previous exit value. Changing IREVCM to any other value between calls will result in an error.

*On intermediate exit:* IREVCM has the following meanings.

- 1 The calling program must compute the matrix-vector product  $y = OPx$ , where  $x$  is stored in X (by default) or in the array COMM (starting from the location given by the first element of ICOMM) when the option **Pointers** = Yes is set in a prior call to F12ADF. The result  $y$  is returned in X (by default) or in the array COMM (starting from the location given by the second element of ICOMM) when the option **Pointers** = Yes is set.
- 1 The calling program must compute the matrix-vector product  $y = OPx$ . This is similar to the case IREVCM = –1 except that the result of the matrix-vector product  $y = Bx$  (as required in some computational modes) has already been computed and is available in MX (by default) or in the array COMM (starting from the location given by the third element of ICOMM) when the option **Pointers** = Yes is set.
- 2 The calling program must compute the matrix-vector product  $y = Bx$ , where  $x$  is stored as described in the case IREVCM = –1 and  $y$  is returned in the location described by the case IREVCM = 1.
- 3 Compute the NSHIFT real and imaginary parts of the shifts where the real parts are to be returned in the first NSHIFT locations of the array X and the imaginary parts are to be returned in the first NSHIFT locations of the array MX. Only complex conjugate pairs of shifts may be applied and the pairs must be placed in consecutive locations. This value of IREVCM will only arise if the option **Supplied Shifts** is set in a prior call to F12ADF which is intended for experienced users only; the default and recommended option is to use exact shifts (see Lehoucq *et al.* (1998) for details).
- 4 Monitoring step: a call to F12AEF can now be made to return the number of Arnoldi iterations, the number of converged Ritz values, their real and imaginary parts, and the corresponding Ritz estimates.

*On final exit:* IREVCM = 5: F12ABF has completed its tasks. The value of IFAIL determines whether the iteration has been successfully completed, or whether errors have been detected. On successful completion F12ACF must be called to return the requested eigenvalues and eigenvectors (and/or Schur vectors).

*Constraint:* on initial entry, IREVCM = 0; on re-entry IREVCM must remain unchanged.

- 2: RESID(\*) – **double precision** array *Input/Output*  
**Note:** the dimension of the array RESID must be at least N.  
*On initial entry:* RESID need not be set unless the option **Initial Residual** has been set in a prior call to F12ADF in which case RESID should contain an initial residual vector, possibly from a previous run.  
*On intermediate re-entry:* RESID must be unchanged from its previous exit. Changing RESID to any other value between calls may result in an error exit.  
*On intermediate exit:* RESID contains the current residual vector.  
*On final exit:* RESID contains the final residual vector.
- 3: V(LDV,\*) – **double precision** array *Input/Output*  
**Note:** the second dimension of the array V must be at least  $\max(1, \text{NCV})$ .  
*On initial entry:* V need not be set.  
*On intermediate re-entry:* V must be unchanged from its previous exit.  
*On intermediate exit:* V contains the current set of Arnoldi basis vectors.  
*On final exit:* V contains the final set of Arnoldi basis vectors.
- 4: LDV – INTEGER *Input*  
*On entry:* the first dimension of the array V as declared in the (sub)program from which F12ABF is called.  
**Constraint:**  $\text{LDV} \geq \text{N}$ .
- 5: X(\*) – **double precision** array *Input/Output*  
**Note:** the dimension of the array X must be at least N if **Pointers** = No (default) and at least 1 if **Pointers** = Yes.  
*On initial entry:* X need not be set.  
*On intermediate re-entry:* if **Pointers** = Yes, X need not be set.  
If **Pointers** = No, X must contain the result of  $y = \text{OP}x$  when IREVCM returns the value -1 or +1. It must return the real parts of the computed shifts when IREVCM returns the value 3.  
*On intermediate exit:* if **Pointers** = Yes, X is not referenced.  
If **Pointers** = No, X contains the vector  $x$  when IREVCM returns the value -1 or +1.  
*On final exit:* X does not contain any useful data.
- 6: MX(\*) – **double precision** array *Input/Output*  
**Note:** the dimension of the array MX must be at least N if **Pointers** = No (default) and at least 1 if **Pointers** = Yes.  
*On initial entry:* MX need not be set.  
*On intermediate re-entry:* if **Pointers** = Yes, MX need not be set.  
If **Pointers** = No, MX must contain the result of  $y = Bx$  when IREVCM returns the value 2. It must return the imaginary parts of the computed shifts when IREVCM returns the value 3.  
*On intermediate exit:* if **Pointers** = Yes, MX is not referenced.  
If **Pointers** = No, MX contains the vector  $Bx$  when IREVCM returns the value +1.  
*On final exit:* MX does not contain any useful data.

- 7:     NSHIFT – INTEGER *Output*  
*On intermediate exit:* if the option **Supplied Shifts** is set and IREVCM returns a value of 3, NSHIFT returns the number of complex shifts required.
- 8:     COMM(\*) – *double precision* array *Communication Array*  
 COMM, on initial entry, must remain unchanged following a call to the setup routine F12AAF.
- 9:     ICOMM(\*) – INTEGER array *Communication Array*  
 ICOMM, on initial entry, must remain unchanged following a call to the setup routine F12AAF.
- 10:    IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On initial entry, the maximum number of iterations  $\leq 0$ , the option **Iteration Limit** has been set to a non-positive value.

IFAIL = 2

The options **Generalized** and **Regular** are incompatible.

IFAIL = 3

The option **Initial Residual** was selected but the starting vector held in RESID is zero.

IFAIL = 4

The maximum number of iterations has been reached. Some Ritz values may have converged; a subsequent call to F12ACF will return the number of converged values and the converged values.

IFAIL = 5

No shifts could be applied during a cycle of the implicitly restarted Arnoldi iteration. One possibility is to increase the size of NCV relative to NEV (see Section 5 of the document for F12AAF for details of these parameters).

IFAIL = 6

Could not build a factorization. Consider changing NCV or NEV in the initialization routine (see Section 5 of the document for F12AAF for details of these parameters).

IFAIL = 7

Unexpected error in internal call to compute eigenvalues and corresponding error bounds of the current upper Hessenberg matrix. Please contact NAG.

IFAIL = 8

Either the initialization routine F12AAF has not been called prior to the first call of this routine or a communication array has become corrupted.

IFAIL = 9

An unexpected error has occurred. Please contact NAG.

## 7 Accuracy

The relative accuracy of a Ritz value,  $\lambda$ , is considered acceptable if its Ritz estimate  $\leq \text{Tolerance} \times |\lambda|$ . The default **Tolerance** used is the *machine precision* given by X02AJF.

## 8 Further Comments

None.

## 9 Example

The example solves  $Ax = \lambda x$  in shift-invert mode, where  $A$  is obtained from the standard central difference discretization of the convection-diffusion operator  $\frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \rho \frac{\partial u}{\partial x}$  on the unit square, with zero Dirichlet boundary conditions. The shift used is a real number.

### 9.1 Program Text

```
*      F12ABF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          LICOMM, NIN, NOUT
      PARAMETER        (LICOMM=140,NIN=5,NOUT=6)
      INTEGER          MAXN, MAXNCV, LDV
      PARAMETER        (MAXN=256,MAXNCV=30,LDV=MAXN)
      INTEGER          LCOMM
      PARAMETER        (LCOMM=3*MAXN+3*MAXNCV*MAXNCV+6*MAXNCV+60)
      INTEGER          IMON
      PARAMETER        (IMON=0)
      DOUBLE PRECISION ONE, TWO
      PARAMETER        (ONE=1.0D+0,TWO=2.0D+0)
*      .. Local Scalars ..
      DOUBLE PRECISION H, RHO, S, S1, S2, S3, SIGMAI, SIGMAR
      INTEGER          I, IFAIL, IFAIL1, INFO, IREVCN, J, N, NCONV, NCV,
+      NEV, NITER, NSHIFT, NX
*      .. Local Arrays ..
      DOUBLE PRECISION COMM(LCOMM), D(MAXNCV,3), DD(MAXN), DL(MAXN),
+      DU(MAXN), DU2(MAXN), MX(MAXN), RESID(MAXN),
+      V(LDV,MAXNCV), X(MAXN)
      INTEGER          ICOMM(LICOMM), IPIV(MAXN)
*      .. External Functions ..
      DOUBLE PRECISION DNRM2
      EXTERNAL          DNRM2
*      .. External Subroutines ..
      EXTERNAL          DGTTRF, DGTTRS, F12AAF, F12ABF, F12ACF, F12ADF,
+      F12AEF
*      .. Intrinsic Functions ..
*
      INTRINSIC          DBLE
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F12ABF Example Program Results'
      WRITE (NOUT,*)
```

```

*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX, NEV, NCV, RHO, SIGMAR, SIGMAI
      N = NX*NX
      IF (N.LT.1 .OR. N.GT.MAXN) THEN
        WRITE (NOUT,99999) 'N is out of range: N = ', N
      ELSE IF (NCV.GT.MAXNCV) THEN
        WRITE (NOUT,99999) 'NCV is out of range: NCV = ', NCV
      ELSE
        IFAIL = 0
        CALL F12AAF(N,NEV,NCV,ICOMM,LICOMM,COMM,LCOMM,IFAIL)
*      Set the mode.
        CALL F12ADF('SHIFTED INVERSE REAL',ICOMM,COMM,IFAIL)
*      Construct C = A - SIGMA*I, and factorize using DGTTRF/F07CDF.
        H = ONE/DBLE(N+1)
        S = RHO*H/TWO
        S1 = -ONE - S
        S2 = TWO - SIGMAR
        S3 = -ONE + S
        DO 20 J = 1, N - 1
          DL(J) = S1
          DD(J) = S2
          DU(J) = S3
20      CONTINUE
        DD(N) = S2
*
        CALL DGTTRF(N,DL,DD,DU,DU2,IPIV,INFO)
*
        IREVCM = 0
        IFAIL = -1
40      CONTINUE
        CALL F12ABF(IREVCM,RESID,V,LDV,X,MX,NSHIFT,COMM,ICOMM,IFAIL)
        IF (IREVCM.NE.5) THEN
          IF (IREVCM.EQ.-1 .OR. IREVCM.EQ.1) THEN
*            Perform  $x \leftarrow \text{inv}[A - \text{SIGMA} \cdot I] \cdot x$  using DGTTRS/F07CEF.
            CALL DGTTRS('N',N,1,DL,DD,DU,DU2,IPIV,X,N,INFO)
          ELSE IF (IREVCM.EQ.4 .AND. IMON.NE.0) THEN
*            Output monitoring information
            CALL F12AEF(NITER,NCONV,D,D(1,2),D(1,3),ICOMM,COMM)
            WRITE (6,99998) NITER, NCONV, DNRM2(NEV,D(1,3),1)
          END IF
          GO TO 40
        END IF
        IF (IFAIL.EQ.0) THEN
*          Post-Process using F12ACF to compute eigenvalues/vectors.
          IFAIL1 = 0
          CALL F12ACF(NCONV,D,D(1,2),V,LDV,SIGMAR,SIGMAI,RESID,V,LDV,
+            COMM,ICOMM,IFAIL1)
*          Print computed eigenvalues.
          WRITE (NOUT,99996) NCONV
          DO 60 I = 1, NCONV
            WRITE (NOUT,99995) I, D(I,1), D(I,2)
60      CONTINUE
        ELSE
          WRITE (NOUT,99997) IFAIL
        END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,'Iteration',1X,I3,',', No. converged =',1X,I3,',', norm o',
+      'f estimates =',E16.8)
99997 FORMAT (1X,' NAG Routine F12ABF Returned with IFAIL = ',I6)
99996 FORMAT (1X,'/ The ',I4,' Ritz values of closest to unity are:',/)
99995 FORMAT (1X,I8,5X,'( ',F12.4,', ',F12.4,', )')
      END

```

## 9.2 Program Data

F12ABF Example Program Data

10 4 20 10.0 1.0 0.0 : Values for NX NEV NCV RHO SIGMAR and SIGMAI

## 9.3 Program Results

F12ABF Example Program Results

The 4 Ritz values of closest to unity are:

1	(	1.0192	,	0.0000	)
2	(	0.9656	,	0.0000	)
3	(	1.0738	,	0.0000	)
4	(	0.9129	,	0.0000	)

---